

## Objectifs de ce module

Je serai capable :

- d'utiliser des fonctions complexes dérivées des fonctions statistiques de base déjà maîtrisées
- de déterminer les données nécessaires à la résolution d'un problème et d'appliquer la fonction adéquate au problème à résoudre par rapport aux données à disposition
- de combiner toutes les fonctions déjà étudiées pour créer des feuilles de calculs complexes

## Introduction

Dans l'apprentissage de l'utilisation d'un tableur, l'apprenant maîtrise relativement vite des fonctions simples, telles que : opérateurs de base + - / \* ^ ; =somme() ; =produit() ; =nb() ; =max () ; =min() ; =moyenne() ; =recherche() ; =recherchev() ; =rechercheh() ; =index() ; =equiv() ; =si() ; etc.

Ces fonctions relativement limitées ne permettent cependant pas toujours l'élaboration de feuilles de calculs complexes propres à résoudre les divers besoins l'entreprise. Il est donc nécessaire de maîtriser les fonctions dérivées des fonctions statistiques de base. Par exemple :

Fonction de base	Dérivée 1	Dérivée 2
<b>=nb()</b>	<b>=nb.si()</b>	<b>=nb.si.ens()</b>
Calcul du nbre total de données	Calcul du nbre de données répondant à un critère	Calcul du nbre de données répondant à plusieurs critères
Ex. : calcul du nbre total des commandes réalisées durant la période	Ex. : calcul du nbre de commandes réalisées par le vendeur Dupont uniquement (1 critère de sélection)	Ex. : calcul du nbre de commandes sur le produit no 1, en Suisse romande, réalisées par le vendeur Dupont (3 critères de sélection)

Ou :

Fonction de base	Dérivée 1	Dérivée 2	Dérivée 3
<b>=somme()</b>	<b>=somme.si()</b>	<b>=somme.si.ens()</b>	<b>=sommeprod()</b>
Addition de valeurs	Addition de valeurs répondant à un critère	Addition de valeurs répondant à plusieurs critères	Addition de valeurs répondant à plusieurs critères (calculs matriciels)
Ex. : calcul (addition) du total des ventes (CAN) enregistrées pour la période	Ex. : calcul du CAN pour le vendeur Dupont uniquement (1 critère de sélection)	Ex. : calcul du CAN sur le produit no 1, écoulé en Suisse romande, par le vendeur Dupont (3 critères de sélection)	Ex. : calcul du CAN sur le produit no 1, écoulé en Suisse romande, par le vendeur Dupont (3 critères de sélection)

La combinaison de ces fonctions permet d'élaborer des feuilles de calculs à la fois complexes et plus souples que l'utilisation d'un rapport de tableau croisé dynamique.

## Exemple d'application

Une entreprise relève les commandes passées par sa force de vente au fur et à mesure qu'elle les reçoit en saisissant les données suivantes : Région, Vendeur, Montant, Date, No de commande, Article.

Elle désire calculer diverses statistiques à partir de ces données collectées, telles que :

- Total du Chiffre d'affaires de la période (selon les prestations convenues)
- Total du Chiffre d'affaires par date
- Total du Chiffre d'affaires sur un produit en particulier, pour un vendeur en particulier
- Total du Chiffre d'affaires sur un produit en particulier, pour un vendeur en particulier, sur une région en particulier

Ou :

- Nbre total de commandes passées
- Nbre de commandes passées par région
- Nbre de commandes passées sur un produit en particulier, pour un vendeur en particulier
- Nbre de commandes passées sur un produit en particulier, pour un vendeur en particulier, sur une région en particulier

Les possibilités sont sans fin et dépendent des besoins de l'entreprise en informations (desk research).

### Méthode de résolution :

Dans une 1<sup>ère</sup> étape, établir un tableur comprenant deux parties distinctes : **I. Données** (matrice de base) et **II. Résultats** (qui se calculent automatiquement en fonction des données saisies par l'utilisateur).

Dans une 2<sup>ème</sup> étape, déterminer quelles données (à saisir) et quelles fonctions permettent d'obtenir les résultats souhaités. Cela dépend de la complexité du problème posé comme par exemple : type de données à disposition, mise en page des données, nbre de critères de sélection des données, etc.

**Attention** : l'utilisation des fonctions dérivées nécessite une bonne compréhension de la mise en page des données et des éléments de la syntaxe de la formule :

- Les données de la matrice doivent être saisies sous forme de listes (cf. page suivante) :
  - étiquettes (champs) en 1<sup>ère</sup> ligne, ce qui donne le nbre de colonnes de la matrice
  - données cohérentes dans chaque colonne, ce qui donne le nbre de lignes
- La syntaxe de la formule et en général la suivante : `SOMME.SI(plage; critère; [somme_plage])`
  - plage 1: référence de la colonne qui contient le critère de sélection
  - critère 1: valeur contenue dans la 1<sup>ère</sup> plage qui sert de critère de sélection
  - somme\_plage : référence de la colonne (plage de base qui contient les valeurs à additionner). Seules les valeurs correspondant au critère de sélection de la plage 1 seront additionnées

- I. Données (exemple de mise en page)

	A	B	C	D	E	F
1	<b>Région</b>	<b>Vendeur</b>	<b>Montant</b>	<b>Date</b>	<b>No Commande</b>	<b>Article</b>
2	Suisse romande	Barbariol	440.00	18.07.12	10248	HB-34
3	Suisse romande	Severino	1'863.40	20.07.12	10249	PZ-4
4	Suisse allemande	Penuchot	1'552.60	22.07.12	10250	HB-34
5	Suisse allemande	Lengel	654.06	17.07.12	10251	PZ-4
6	Suisse allemande	Penuchot	3'597.90	19.07.12	10252	BC-12
7	Suisse allemande	Lengel	1'444.80	21.07.12	10253	PZ-4
8	Suisse romande	Barbariol	556.62	16.07.12	10254	HB-34
9	Suisse romande	Delgado	2'490.50	18.07.12	10255	PZ-4
10	Suisse allemande	Lengel	517.80	20.07.12	10256	HB-34
11	Suisse allemande	Penuchot	1'119.90	22.07.12	10257	PZ-4
12	Suisse allemande	Delaney	1'614.88	24.07.12	10258	PZ-4
13	Suisse allemande	Penuchot	100.80	19.07.12	10259	AX-16
14	Suisse allemande	Penuchot	1'504.65	21.07.12	10260	PZ-4
15	Suisse allemande	Penuchot	448.00	23.07.12	10261	HB-34
16	Suisse allemande	Caron	584.00	18.07.12	10262	HB-34
17	Suisse romande	Delgado	1'873.80	20.07.12	10263	PZ-4

- II. Résultats (exemple de mise en page)

	A	B	C	D	E	F
20	<b>Exemples de statistiques sur la période</b>				données	formule
21						
22	Total du CAN de la période sous revue					20'363.71
23	Total des commandes réalisées durant la période sous revue					16
24	Total des commandes passées par le vendeur sous revue (à saisir)				Penuchot	6
25	Total du CAN du vendeur sous revue					8'323.85
26	Total du CAN du vendeur sous revue pour l'article (à saisir)				HB-34	2'000.60
27						
28	Avec les formules :					
29						
30	Total du CAN de la période sous revue					20'363.71
32	Pas de critère de sélection				=SOMME(\$C\$2:\$C\$17)	
34	Total des commandes passées durant la période sous revue					16
36	Pas de critère de sélection				=NB(\$C\$2:\$C\$17)	
38	Total des commandes passées par le vendeur sous revue (à saisir)				Penuchot	6
40	Un critère (la plage 1 contient le critère 1) : (Plage 1;Critère 1)				=NB.SI(\$B\$2:\$B\$17;E\$38)	
41						
42	Total du CAN du vendeur sous revue					8'323.85
44	Un critère (page 2 = valeurs à additionner) : (Plage 1;Critère 1;Plage 2)				=SOMME.SI(\$B\$2:\$B\$17;E\$38;\$C\$2:\$C\$17)	
45						
46	Total du CAN du vendeur sous revue pour l'article (à saisir)				HB-34	2'000.60
48	Plusieurs critères *				=SOMME.SI.ENS(\$C\$2:\$C\$17;\$B\$2:\$B\$17;E\$38;\$F\$2:\$F\$17;E\$46)	
50	(*Plage de base (valeurs à additionner) ; Plage 1;Critère 1 ; Plage 2;Critère 2 ; etc.)					

Remarque : la fonction complexe =sommeprod() peut remplacer les fonctions =somme.si.ens() ou =nb.si.ens()

L'exercice No 1 permet de s'entraîner.

## Pour aller plus loin

Parmi les fonctions élaborées proposées par le tableur figurent celles de type : base de données ou BD. Ces fonctions sont des dérivées des fonctions statistiques déjà maîtrisées et permettent de se familiariser avec l'utilisation de programme de gestions de données (1<sup>ère</sup> approche).

Fonction de base	Dérivée BD	Fonction de base	Dérivée BD
=somme()	=bdsomme()	=min()	=bdmin()
=moyenne()	=bdmoyenne()	=nb()	=bdnb()
=max()	=bdmax()	=recherche() et ses variantes	=bdlire()

**Attention** : un tableur est un outil de calculs et doit le rester. Il peut cependant être utilisé grâce à sa souplesse par rapport aux programmes de gestion intégrés (ERP en anglais – type SAP ou autre) pour des problèmes de gestions de données simples. Ses limites sont par contre vite atteintes : il faut donc garder à l'esprit qu'à tout problème correspond un outil adapté.

## Exemple d'application

Les formules BD se prêtent particulièrement bien lorsque qu'une entreprise doit déterminer une valeur par rapport à plusieurs variables (en ce sens, ces fonctions remplacent avantageusement les fonctions =EQUIV() et =INDEX() déjà étudiée).

Par exemple :

- Dans une station de ski, le prix du forfait peut dépendre de 2 données : l'année de naissance du skieur et la durée du forfait (1 jour, 2 jours, etc.)
- Dans une agence de location de voitures, le prix de location peut dépendre de la catégorie du véhicule loué et de la durée de la location
- La facture d'un opérateur téléphonique peut dépendre du type d'abonnement choisi par le client et de la durée ou de la quantité des prestations consommées (durée de l'appel inclus dans l'abonnement, nbre de Mo ou de SMS gratuits, en plus, etc.)

### Méthode de résolution :

Dans une 1<sup>ère</sup> étape, établir un tableur comprenant deux parties distinctes : **I. Données** (matrice de base) et **II. Résultats** (qui se calculent automatiquement en fonction des données saisies par l'utilisateur).

Dans une 2<sup>ème</sup> étape, déterminer quelles données (à saisir) et quelles fonctions permettent d'obtenir les résultats souhaités. Cela dépend de la complexité du problème posé comme par exemple : type de données à disposition, mise en page des données, nbre de critères de sélection des données, etc.

**Attention** : l'utilisation des fonctions de type BD nécessite une bonne compréhension de la mise en page des données et des éléments de la syntaxe de la formule :

- **I. Données** : matrice (base de données) saisie sous forme de listes :
  - étiquettes (champs) en 1<sup>ère</sup> ligne, ce qui donne le nbre de colonnes de la matrice et
  - données cohérentes dans chaque colonne, ce qui donne le nbre de lignes.

◇	A	B	C	D	E	F
1						
2	<b>Tarifs des forfaits de ski - dégressif selon la durée</b>					
3						
4				Non modifiable	Données	Calculs
5						
6	Jour	Adulte	Senior	Jeune	Enfant	Bambini
7	1	73	67	61	37	0
8	2	135	125	115	68	0
9	3	196	181	166	98	0
10	4	253	234	215	127	0
11	5	309	285	261	155	0
12	6	359	331	303	180	0
13	7	404	372	340	202	0

- La syntaxe de la formule et la suivante : **BDLIRE(base\_de\_données; champ; critères)**
  - base de données : références absolues de la matrice, étiquettes et données
  - champ : nom de l'étiquette (colonne) dont on cherche une des valeurs
  - critères : nom de l'étiquette (colonne) et valeur dans cette colonne qui sert de critère
- **II. Résultats** : on demande au client son année de naissance et la durée de son forfait pour calculer automatiquement le tarif applicable

◇	A	B	C	D
24				
25	Année de naissance	Jour	Statut (champ)	Tarif
26	2006	3	Enfant	98
27				
28	D26 = BDLIRE(\$A\$6:\$F\$13;C26;B25:B26)			
29				

- les cellules A6:F13 = adresse de la base de données (cf. ci-dessus)
- la cellule C26 = nom de l'étiquette (colonne) dont on cherche une des valeurs : on cherche le tarif « Enfant »
- les cellules B25:B26 = nom de l'étiquette (colonne) et valeur dans cette colonne qui sert de critère : on cherche le tarif pour « 3 jours » (attention : disposition l'une sur l'autre à conserver)

L'exercice No 2 permet de s'entraîner.